Numerical Solution of the Time-Dependent Schrödinger Equation

Wolfgang Lösch

Introduction

This article describes a numerical approach to solve the time-dependent SCHRÖDINGER equation of a particle in a one-dimensional potential. The work was inspired by a publication of Goldberg, Schey und Schwartz [1]. The structure of the article is based on the chapters of the books by Schmid, Spitz and Lösch [2][3].

Problem

An α-particle (helium nucleus) starts at position x_M at time $t = 0$ and moves towards a potential barrier or well of width x_V. The α-particle is considered as an elementary particle with no internal degrees of freedom. The potential function is

$$V(x) = \begin{cases} \pm E_{\text{pot}} & \text{for } \; x_{\text{pot}} - x_V/2 \; \leq \; x \; \leq \; x_{\text{pot}} + x_V/2 \\ 0 & \text{otherwise} \end{cases} \tag{1}$$

with suitable values for E_{pot} and x_{pot}. The movement of the α-particle is in the direction of increasing values of x and it is assumed that $x_M < x_{\text{pot}} - x_V/2$ at $t = 0$.

To treat this problem in quantum mechanics we are looking for a solution of the time-dependent SCHRÖDINGER equation

$$i\hbar \frac{\partial \psi}{\partial t}(x,t) = H\psi(x,t) \tag{2}$$

with the HAMILTONian

$$H = -\frac{\hbar^2}{2m}\frac{\partial^2}{\partial x^2} + V(x). \tag{3}$$

Equation (2) is a first-order differential equation with respect to time. Therefore, the wave function $\psi(x,t)$ can uniquely be calculated from the initial value of the wave function $\psi(x,0)$ at all times t. So we are looking for a solution $\psi(x,t)$ of the SCHRÖDINGER equation (2), determined by a physically meaningful initial condition $\psi(x,0)$.

Let us first consider a free particle, where $V = 0$. In this case the SCHRÖDINGER equation (2) reduces to

$$i\hbar \frac{\partial \psi}{\partial t}(x,t) = -\frac{\hbar^2}{2m}\frac{\partial^2 \psi}{\partial x^2}(x,t). \tag{4}$$

Particular solutions of equation (4) are the (non-normalizable) plane waves

$$\psi_k(x,t) = \frac{1}{\sqrt{2\pi}} \exp\{i(kx - \omega_k t)\}, \tag{5}$$

where

$$\omega_k = \frac{\hbar k^2}{2m} \tag{6}$$

with an arbitrary wave number k.

Other solutions of equation (4) are obtained by superposition of plane waves with any amplitude $A(k)$, provided the integrals exist:

$$\psi(x,t) = \int_{-\infty}^{\infty} A(k)\, \psi_k(x,t)\, dk = \frac{1}{\sqrt{2\pi}} \int_{-\infty}^{\infty} A(k) \exp\left\{i(kx - \omega_k t)\right\} dk\,. \tag{7}$$

Inserting $t = 0$ in equation (7), we obtain the initial condition

$$\psi(x,0) = \frac{1}{\sqrt{2\pi}} \int_{-\infty}^{\infty} A(k) \exp\left(ikx\right) dk. \tag{8}$$

We immediately recognize this is a FOURIER transformation. The function $A(k)$ is obtained by the inverse transformation

$$A(k) = \frac{1}{\sqrt{2\pi}} \int_{-\infty}^{\infty} \psi(x,0) \exp\left(-ikx\right) dx. \tag{9}$$

Now we have to choose a suitable term for $A(k)$. Different approaches are possible to describe the motion of the α-particle. Only a wave function $\psi\,(x,t)$ which is square integrable and thus normalizable may give us a physically meaningful solution. A common approach used in many textbooks on quantum mechanics is

$$A(k) = \sqrt{\frac{\sigma}{\sqrt{\pi}}} \exp\left\{\frac{-\sigma^2(k-k_M)^2}{2}\right\} \exp\left\{-i(k-k_M)x_M\right\}. \tag{10}$$

Inserting this into equation (8), we obtain the initial condition

$$\psi(x,0) = \frac{1}{\sqrt{\sigma\sqrt{\pi}}} \exp\left\{\frac{-(x-x_M)^2}{2\sigma^2}\right\} \exp\left\{ik_M x\right\}, \tag{11}$$

leading to the probability density

$$|\psi(x,0)|^2 = \frac{1}{\sigma\sqrt{\pi}} \exp\left\{\frac{-(x-x_M)^2}{\sigma^2}\right\}. \tag{12}$$

This is the famous GAUSSian bell curve, also known as GAUSSian wave packet, where x_M specifies the peak and σ specifies the width.

$\psi(x,0)$ is normalized by the factor $1/\sqrt{\sigma\sqrt{\pi}}$, such that

$$\int_{-\infty}^{\infty} |\psi(x,0)|^2\, dx = \frac{1}{\sigma\sqrt{\pi}} \int_{-\infty}^{\infty} \exp\left\{\frac{-(x-x_M)^2}{\sigma^2}\right\} dx = 1. \tag{13}$$

The term $\exp\left(ik_M x\right)$ causes the wave packet to move in the direction of increasing x values.

Now let us have a look at the expectation values and the standard deviations of the position operator x and the momentum operator $p = \hbar/i\ \partial/\partial x$, as well as the expectation value of the HAMILTONian

$$H = \frac{p^2}{2m} = -\frac{\hbar^2}{2m}\frac{\partial^2}{\partial x^2} \tag{14}$$

at $t = 0$.

Let us recall the definitions of the expectation value $\langle O\rangle_t$ and the standard or root-mean-square deviation $(\Delta O)_t$ of an operator O:

$$\langle O\rangle_t = \int_{-\infty}^{\infty} \overline{\psi(x,t)}\, O\psi(x,t)\, dx, \tag{15}$$

$$(\Delta O)_t = \sqrt{\langle (O - \langle O\rangle_t)^2\rangle_t} = \sqrt{\langle O^2\rangle_t - \langle O\rangle_t^2}. \tag{16}$$

For the GAUSSian wave packet (11) we obtain the follwing results:

$$\langle x\rangle_0 = x_M, \tag{17}$$

$$\langle p\rangle_0 = \hbar k_M, \tag{18}$$

$$(\Delta x)_0 = \frac{\sigma}{\sqrt{2}}, \tag{19}$$

$$(\Delta p)_0 = \frac{\hbar}{\sigma\sqrt{2}}, \tag{20}$$

$$E = \langle H\rangle_0 = \frac{\langle p^2\rangle_0}{2m} = \frac{\langle p\rangle_0^2 + (\Delta p)_0^2}{2m} = \frac{\hbar^2 k_M^2}{2m} + \frac{(\Delta p)_0^2}{2m}. \tag{21}$$

Equation (17) corresponds to the classical idea of having the α-particle started at position x_M. However, due to equation (19) we have to accept a position uncertainty $(\Delta x)_0$ in the order of σ.

Equation (21) shows a typical property of quantum mechanics. The relation $E_{\text{cl}} = p_{\text{cl}}^2/2m$, well-known from classical mechanics, can only be obtained by setting $(\Delta p)_0 = 0$. According to the HEISENBERG uncertainty principle

$$(\Delta x)_t(\Delta p)_t \geq \frac{\hbar}{2}, \tag{22}$$

this implies that the position is uncertain and the α-particle could be found anywhere. But since we want to locate the α-particle in a given area, we have to accept a momentum uncertainty $(\Delta p)_0$. However, this causes the wave packet to spread out with progressing time, which makes it difficult to localize the particle. So we must try to keep the position uncertainty $(\Delta x)_0$ as small as possible at $t = 0$.

From equation (19) it can be seen, that a small position uncertainty requires a small value for σ. On the other hand we have to ensure that $(\Delta p)_0$ according to equation (20) may not get too large.

Let us try $\sigma = 5\,\text{fm}$. With $\hbar = 6.4655\,\text{fm}\sqrt{\text{MeVu}}$ we then obtain $(\Delta p)_0 \approx 0.9\,\sqrt{\text{MeVu}}$ and $(\Delta p)_0^2/2m \approx 0.1\,\text{MeV}$ for an α-particle with mass $m \approx 4\text{u}$. If we choose energy values $E \geq 10\,\text{MeV}$ we can neglect the term $(\Delta p)_0^2/2m$ and thus write down equation (21) in the following form:

$$k_M = \frac{\sqrt{2mE}}{\hbar}. \tag{23}$$

Let us now return to our original problem and look at the situation in the presence of a potential according to equation (1). Our previous considerations remain valid if we choose the start position x_M at $t = 0$ in such a way that the wave packet is far enough from the potential barrier or well, with a wave function $\psi(x,t)$ practically vanishing for $x \geq x_{\text{pot}} - x_V/2$. A distance of about 20 fm can be considered to be sufficient for a width of 5 fm.

Numerical Method

The formal solution of the time-dependent SCHRÖDINGER equation (2) is

$$\psi(x,t) = \exp\left\{-\frac{i}{\hbar}Ht\right\}\psi(x,0)\,, \tag{24}$$

with the exponential operator

$$\exp\left\{-\frac{i}{\hbar}Ht\right\} = \sum_{l=0}^{\infty}\frac{\left(-\frac{i}{\hbar}Ht\right)^l}{l\,!} = 1 - \frac{i}{\hbar}Ht - \frac{1}{2\hbar^2}H^2t^2 + \frac{i}{6\hbar^3}H^3t^3 + \ldots\,, \tag{25}$$

defined in terms of a TAYLOR series.

The exponential operator $\exp\{-iHt/\hbar\}$ is unitary, and thus the norm of the wave function does not change over time. We are now looking for a unitary approximation of this operator to bring equation (24) in a form suitable for numerical evaluation. Unfortunately the approximation

$$\exp\left\{-\frac{i}{\hbar}Ht\right\} \approx \sum_{l=0}^{L}\frac{\left(-\frac{i}{\hbar}Ht\right)^l}{l\,!} \tag{26}$$

is not unitary for finite values of L.

The CAYLEY form

$$C = \left(1 - \frac{i}{2\hbar}Ht\right)\left(1 + \frac{i}{2\hbar}Ht\right)^{-1}\,, \tag{27}$$

however, is unitary (see reference [4]), and the expansion is

$$C = 1 - \frac{i}{\hbar}Ht - \frac{1}{2\hbar^2}H^2t^2 + \frac{i}{4\hbar^3}H^3t^3 + \ldots\,. \tag{28}$$

Comparing this to equation (25), we can see that this is a good approximation of $\exp\{-iHt/\hbar\}$, which is exact up to the term of order t^2. We can now write equation (24) as

$$\psi(x,t) = C\psi(x,0). \tag{29}$$

The operators $1-iHt/2\hbar$ and $1+iHt/2\hbar$ are commutable and from equation (29) we obtain

$$\left(1+\frac{i}{2\hbar}Ht\right)\psi(x,t) = \left(1-\frac{i}{2\hbar}Ht\right)\psi(x,0). \tag{30}$$

To eliminate $\psi(x,t)$ we define the function

$$\phi(x,t) = \psi(x,t) + \psi(x,0)\,, \tag{31}$$

which leads us to the equation

$$\left(\frac{1}{2}+\frac{i}{4\hbar}Ht\right)\phi(x,t) = \psi(x,0)\,. \tag{32}$$

Replacing H by equation (3) we finally get the following differential equation

$$\left(\frac{1}{2}-\frac{i\hbar t}{8m}\frac{\partial^2}{\partial x^2}+\frac{iV(x)t}{4\hbar}\right)\phi(x,t) = \psi(x,0)\,. \tag{33}$$

We now want to derive a *difference* equation from this *differential* equation by discretizing the position x. We apply the three-point formula (see references [2][3])

$$\frac{\partial^2\phi}{\partial x^2}(x,t) = \frac{\phi(x-s,t)-2\phi(x,t)+\phi(x+s,t)}{s^2}\,, \tag{34}$$

where s is a small step and from equation (33) we obtain

$$-\frac{i\hbar t}{8ms^2}\phi(x-s,t)+\left(\frac{1}{2}+\frac{i\hbar t}{4ms^2}+\frac{iV(x)t}{4\hbar}\right)\phi(x,t)-\frac{i\hbar t}{8ms^2}\phi(x+s,t) = \psi(x,0)\,. \tag{35}$$

Defining

$$x_j = js,\;\; \phi_j^t = \phi(x_j,t),\;\; V_j = V(x_j),\;\; \psi_j^t = \psi(x_j,t)\,, \tag{36}$$

we finally get the difference equation

$$-\frac{i\hbar t}{8ms^2}\phi_{j-1}^t+\left(\frac{1}{2}+\frac{i\hbar t}{4ms^2}+\frac{iV_jt}{4\hbar}\right)\phi_j^t-\frac{i\hbar t}{8ms^2}\phi_{j+1}^t = \psi_j^0\,. \tag{37}$$

By discretizing equation (31) we get the desired solution

$$\psi_j^t = \phi_j^t - \psi_j^0\,, \tag{38}$$

where ϕ_j^t is calculated according to equation (37) and ψ_j^0 is set as initial condition.

One aspect we have not yet taken into account. In fact, we have to solve the SCHRÖDINGER equation (2) on the spatial interval $(-\infty, \infty)$. This is not possible for any real computer. Therefore, we restrict ourselves to the interval $[0, b]$ with a suitable upper boundary

$$b = j_b s \tag{39}$$

and we set

$$\psi(x,t) = 0 \quad \text{for} \quad x < 0 \quad \text{and} \quad x > b\,, \tag{40}$$

or in discretized form

$$\psi_j^t = 0 \quad \text{for} \quad j < 0 \quad \text{and} \quad j > j_b\,. \tag{41}$$

For $\phi(x,t)$ we obtain the boundary conditions

$$\phi_j^t = 0 \quad \text{for} \quad j < 0 \quad \text{and} \quad j > j_b\,. \tag{42}$$

Defining

$$c = -\frac{i\hbar t}{8ms^2} \tag{43}$$

and

$$d_j = \frac{1}{2} + \frac{i\hbar t}{4ms^2} + \frac{iV_j t}{4\hbar}, \quad j = 0, \ldots, j_b\,, \tag{44}$$

equation (37) together with the boundary conditions (42) can be written in the following form

$$\begin{pmatrix} d_0 & c & & & 0 \\ c & d_1 & c & & \\ & \ddots & \ddots & \ddots & \\ & & c & d_{j_b-1} & c \\ 0 & & & c & d_{j_b} \end{pmatrix} \begin{pmatrix} \phi_0^t \\ \phi_1^t \\ \vdots \\ \phi_{j_b-1}^t \\ \phi_{j_b}^t \end{pmatrix} = \begin{pmatrix} \psi_0^0 \\ \psi_1^0 \\ \vdots \\ \psi_{j_b-1}^0 \\ \psi_{j_b}^0 \end{pmatrix}. \tag{45}$$

This is a system of linear equations with a tridiagonal matrix. We will use the GAUSSian method to solve these equations. The basic idea of this method is to decompose the tridiagonal matrix into a product of an upper and a lower triangular band matrix. Let us try the following decomposition

$$\begin{aligned} &\begin{pmatrix} d_0 & c & & & 0 \\ c & d_1 & c & & \\ & \ddots & \ddots & \ddots & \\ & & c & d_{j_b-1} & c \\ 0 & & & c & d_{j_b} \end{pmatrix} \\ &\quad = \begin{pmatrix} \alpha_0 & & & & 0 \\ \beta_1 & \alpha_1 & & & \\ & \ddots & \ddots & & \\ & & \beta_{j_b-1} & \alpha_{j_b-1} & \\ 0 & & & \beta_{j_b} & \alpha_{j_b} \end{pmatrix} \begin{pmatrix} 1 & \gamma_0 & & & 0 \\ & 1 & \gamma_1 & & \\ & & \ddots & \ddots & \\ & & & 1 & \gamma_{j_b-1} \\ 0 & & & & 1 \end{pmatrix} \end{aligned} \tag{46}$$

$$
= \begin{pmatrix}
\alpha_0 & \alpha_0\gamma_0 & & & 0 \\
\beta_1 & \beta_1\gamma_0+\alpha_1 & \alpha_1\gamma_1 & & \\
& \ddots & \ddots & \ddots & \\
& & \beta_{j_b-1} & \beta_{j_b-1}\gamma_{j_b-2}+\alpha_{j_b-1} & \alpha_{j_b-1}\gamma_{j_b-1} \\
0 & & & \beta_{j_b} & \beta_{j_b}\gamma_{j_b-1}+\alpha_{j_b}
\end{pmatrix} .
$$

Equating the corresponding coefficients results in a system of recursive formulae for the matrix elements α_j and γ_j

$$
\alpha_0 = d_0, \tag{47}
$$

$$
\gamma_j = \frac{c}{\alpha_j}, \quad j = 0, \dots, j_b - 1\,, \tag{48}
$$

$$
\alpha_{j+1} = d_{j+1} - c\gamma_j, \quad j = 0, \dots, j_b - 1\,. \tag{49}
$$

If we write the right hand side of equation (45) in the following form

$$
\begin{pmatrix} \psi_0^0 \\ \psi_1^0 \\ \vdots \\ \psi_{j_b-1}^0 \\ \psi_{j_b}^0 \end{pmatrix}
= \begin{pmatrix}
\alpha_0 & & & & 0 \\
\beta_1 & \alpha_1 & & & \\
& \ddots & \ddots & & \\
& & \beta_{j_b-1} & \alpha_{j_b-1} & \\
0 & & & \beta_{j_b} & \alpha_{j_b}
\end{pmatrix}
\begin{pmatrix} g_0 \\ g_1 \\ \vdots \\ g_{j_b-1} \\ g_{j_b} \end{pmatrix}
= \begin{pmatrix} \alpha_0 g_0 \\ \beta_1 g_0 + \alpha_1 g_1 \\ \vdots \\ \beta_{j_b-1} g_{j_b-2} + \alpha_{j_b-1} g_{j_b-1} \\ \beta_{j_b} g_{j_b-1} + \alpha_{j_b} g_{j_b} \end{pmatrix} , \tag{50}
$$

then, by equating the corresponding coefficients we obtain the recursive formula

$$
g_0 = \frac{\psi_0^0}{\alpha_0}, \tag{51}
$$

$$
g_j = \frac{\psi_j^0 - c g_{j-1}}{\alpha_j}, \quad j = 1, \dots, j_b\,. \tag{52}
$$

Inserting equations (46) and (50) in equation (45) results in another system of linear equations

$$
\begin{pmatrix}
1 & \gamma_0 & & & 0 \\
& 1 & \gamma_1 & & \\
& & \ddots & \ddots & \\
& & & 1 & \gamma_{j_b-1} \\
0 & & & & 1
\end{pmatrix}
\begin{pmatrix} \phi_0^t \\ \phi_1^t \\ \vdots \\ \phi_{j_b-1}^t \\ \phi_{j_b}^t \end{pmatrix}
= \begin{pmatrix} \phi_0^t + \gamma_0 \phi_1^t \\ \phi_1^t + \gamma_1 \phi_2^t \\ \vdots \\ \phi_{j_b-1}^t + \gamma_{j_b-1} \phi_{j_b}^t \\ \phi_{j_b}^t \end{pmatrix}
= \begin{pmatrix} g_0 \\ g_1 \\ \vdots \\ g_{j_b-1} \\ g_{j_b} \end{pmatrix} . \tag{53}
$$

By equating the corresponding coefficients we obtain the recursive formula

$$
\phi_{j_b}^t = g_{j_b}, \tag{54}
$$

$$
\phi_j^t = g_j - \gamma_j \phi_{j+1}^t, \quad j = j_b - 1, \dots, 0\,, \tag{55}
$$

which finally yields the desired solution of equation (45).

How do we now proceed in practice? Starting with the initial condition $\psi(x_j, 0) = \psi_j^0$, $j = 0, 1, 2, \ldots, j_b$, we compute $\psi(x_j, 0) = \psi_j^0$, $j = 0, 1, 2, \ldots, j_b$ on the basis of equation (45) and equation (38) by applying the Gaussian method explained above.

Then, this solution is set as the new initial condition and we have to solve equation (45) again. This second run results in the solution $\psi(x_j, 2t) = \psi_j^{2t}$, $j = 0, 1, 2, \ldots, j_b$, which, as before, has to be set as the new initial condition for the third run.

By repeating this procedure we successively obtain all the solutions $\psi(x_j, t)$, $\psi(x_j, 2t)$, $\psi(x_j, 3t)$, $\ldots$, $\psi(x_j, T)$ up to the time $T = nt$, $n \geq 1$.

Keeping in mind that the approximation (28) is exact up to the term of order t^2 and the approximation (34) only up to the term of order s, we have to choose the time step t and the spatial step s sufficiently small.

Programming

Programming is done in FORTRAN 77. This language is particularly well suited to convert physical quantities and relationships into program code. Figure 1 shows the parameters, variables and arrays corresponding to the physical quantities.

Physical notation	FORTRAN notation	Physical notation	FORTRAN notation
π	`PI`	$V(x_j)$	`V(J)`
σ	`SIGMA`	$\hbar$	`HBAR`
$1/\sqrt{\sigma\sqrt{\pi}}$	`SIGPI`	$\hbar^2/2m$	`H2M`
E	`E`	m	`M`
k_M	`KM`	$\langle x \rangle_t$	`XEXP`
x_M	`XM`	x_j	`X(J)`
b	`B`	d_j	`D(J)`
x_V	`XV`	c	`C`
j_b	`JB`	$\phi(x_j, t)$	`PHI(J)`
s	`S`	$\psi(x_j, t)$	`PSI(J)`
T	`TR`	$\lvert\psi(x_j, t)\rvert^2$	`PSIABS(J)`
n_T	`NT`	α_j	`ALPHA(J)`
t	`T`	γ_j	`GAMMA(J)`
n	`N`	g_j	`G(J)`
E_{pot}	`NV*EPOT` (`NV=-1`, `0` oder `1`)	δ	`DELTA`

Figure 1: Notation

The numerical part of the program is structured as follows:

```
      PROGRAM QM
      INTEGER JMAX, J, JB, NT, N, NV
      REAL HBAR, H2M, M, SIGMA, B, XM, EPOT, EPS, PI
      REAL KM, E, SIGPI, S, TR, T, XEXP, DELTA, XV
      PARAMETER (EPS=0.0005, PI=3.1415926535, JMAX=3000)
      PARAMETER (HBAR=6.4655, H2M=5.1875, M=0.5*HBAR*HBAR/H2M)
      PARAMETER (SIGMA=5., B=20.*SIGMA, XM=B/4., EPOT=20.)
      REAL X(0:JMAX), V(0:JMAX), PSIABS(0:JMAX)
      COMPLEX C, D(0:JMAX), PHI(0:JMAX), PSI(0:JMAX)
      COMPLEX GAMMA(0:JMAX), ALPHA(0:JMAX), G(0:JMAX)

* (Input: NV,XV,E)

      SIGPI=1./(SQRT(SQRT(PI)*SIGMA))
      KM=SQRT(2.*M*E)/HBAR
      TR=M*B/(2.*HBAR*KM)

* (Input: JB,NT)

      S=B/JB
      T=TR/NT
      DO 10 J=0,JB
        X(J)=J*S
        IF(ABS(X(J)-B/2.).LE.XV/2.) THEN
          V(J)=NV*EPOT
        ELSE
          V(J)=0.
        ENDIF
        D(J)=CMPLX(0.5,HBAR*T/(4.*M*S*S)+V(J)*T/(4.*HBAR))
        PSI(J)=SIGPI*CEXP(CMPLX(-0.5*((X(J)-XM)/SIGMA)**2,KM*X(J)))
   10 CONTINUE
      C=CMPLX(0.,-HBAR*T/(8.*M*S*S))

* (Output: PSIABS, XEXP at time t=0)

      N=1
   20 CONTINUE
      IF(N.EQ.1) THEN
        ALPHA(0)=D(0)
        DO 30 J=0,JB-1
          GAMMA(J)=C/ALPHA(J)
          ALPHA(J+1)=D(J+1)-C*GAMMA(J)
   30   CONTINUE
      ENDIF
      G(0)=PSI(0)/ALPHA(0)
```

```
37       DO 40 J=1,JB
38         G(J)=(PSI(J)-C*G(J-1))/ALPHA(J)
39 40    CONTINUE
40       PHI(JB)=G(JB)
41       DO 50 J=JB-1,0,-1
42         PHI(J)=G(J)-GAMMA(J)*PHI(J+1)
43 50    CONTINUE
44       DO 60 J=0,JB
45         PSI(J)=PHI(J)-PSI(J)
46         PSIABS(J)=PSI(J)*CONJG(PSI(J))
47 60    CONTINUE
48       XEXP=S/2.*(X(0)*PSIABS(0)+X(JB)*PSIABS(JB))
49       DO 70 J=1,JB-1
50         XEXP=XEXP+S*X(J)*PSIABS(J)
51 70    CONTINUE

   * (Output: PSIABS, XEXP, N*T)

52       N=N+1
53       IF(((NV.NE.0).AND.(ABS(B/2.-XEXP).LT.(B/2.-XM)).AND.
54      &  (PSIABS(JB/50).LT.EPS).AND.(PSIABS(JB-JB/50).LT.EPS)).OR.
55      &  ((NV.EQ.0).AND.(N.LE.NT))) THEN
56         GOTO 20
57       ELSEIF(NV.EQ.0) THEN
58         DELTA=1.+(HBAR*TR/(M*SIGMA**2))**2
59         DO 80 J=0,JB
60           PSIABS(J)=SIGPI*SIGPI/SQRT(DELTA)
61      &     *REAL(DEXP(DBLE(-((X(J)-XM-HBAR*KM/M*TR)/SIGMA)**2/DELTA)))
62 80      CONTINUE

   * (Output: PSIABS for an analytical wave packet)

63       ENDIF
64       END
```

As an upper limit for the spatial interval $[0, b]$, we choose $b = 100\,\mathrm{fm}$, which results in the $j_b + 1$ nodes $x_j = js$, $j = 0, 1, \ldots, j_b$, with $s = b/j_b$ (lines 14 and 17). The potential barrier or well of width x_V is located in the center of this interval. The height of the barrier or the depth of the well is $E_{\mathrm{pot}} = \pm 20\,\mathrm{MeV}$. The wave packet with energy E starts at $x_M = b/4 = 25\,\mathrm{fm}$. We have to type in values for j_b, x_V und E, whereas E_{pot}, b and x_M are preset (line 7). Other preset quantities are $\hbar = 6.4655\,\mathrm{fm}\sqrt{\mathrm{MeVu}}$, $\hbar^2/2m = \frac{1}{2} \cdot 10.375\,\mathrm{MeVfm}^2 = 5.1875\,\mathrm{MeVfm}^2$ (see references [2][3]), $m = \frac{1}{2}\hbar^2/(\hbar^2/2m)$ and $\sigma = 5\,\mathrm{fm}$ (lines 6 und 7).

What we need now is a reasonable value for the time step t. We look at the expectation value of position of the free wave packet moving from $x_M = b/4$ to $b - x_M = 3b/4$. The time needed for this distance will be used as *reference time* T, such that

$$\langle x \rangle_T = b - x_M. \tag{56}$$

Integrating equation (7), we obtain the wave function

$$\psi(x,T) = \sqrt{\frac{\frac{\sigma}{\sqrt{\pi}}}{\sigma^2 + \frac{i\hbar T}{m}}}\ \exp\left\{-\frac{\sigma^2 k_M^2}{2}\right\} \exp\left\{\frac{(\sigma^2 k_M + i(x-x_M))^2}{2\left(\sigma^2 + \frac{i\hbar T}{m}\right)}\right\} \exp\left(ik_M x_M\right), \tag{57}$$

and using the abbreviation

$$\delta = 1 + \frac{\hbar^2 T^2}{m^2 \sigma^4} \tag{58}$$

we get the probability density

$$|\psi(x,T)|^2 = \frac{1}{\sigma\sqrt{\pi\delta}}\ \exp\left\{\frac{-\left(x - x_M - \frac{\hbar k_M}{m}T\right)^2}{\sigma^2\delta}\right\}. \tag{59}$$

Now we can calculate the expectation value of position according to equation (15)

$$\langle x \rangle_T = x_M + \frac{\hbar k_M T}{m}\,, \tag{60}$$

and with $x_M = b/4$ we obtain the reference time (line 13)

$$T = \frac{mb}{2\hbar k_M}\,. \tag{61}$$

We decompose the time interval $[0,T]$ in n_T subintervals and finally get the time step (line 15)

$$t = \frac{T}{n_T} = \frac{mb}{2\hbar k_M n_T}\,. \tag{62}$$

To keep the program flexible with regard to the time step, the value for n_T has to be typed in.

The potential values $V(x_j)$ are calculated within an `IF` construct (lines 18 to 22). We have three choices: No potential at all (`NV=0`), potential barrier (`NV=1`), or potential well (`NV=-1`). The calculation of the diagonal elements d_j and the off-diagonal element c according to equations (44) and (43) is done in lines 23 and 26, respectively. With k_M given by equation (23) and the normalization factor $1/\sqrt{\sigma\sqrt{\pi}}$ (lines 11 and 12), the initial values $\psi(x_j, 0)$ are calculated in line 24 according to equation (11).

After this preparatory work we are now able to solve the system of linear equations (45). We set the counting index for the number of time steps on $n = 1$ (line 27). Then, the processing of several formulae can be carried out. Since the lines of code speak for themselves, we simply note the cross-references: (47) in line 30, (48) and (49) in the `DO` loop from line 31 to 34, (51) in line 36, (52) in the DO loop from line 37 to 39, (54) in line 40, (55) in the DO loop from line 41 to 43 and (38) in the DO loop from line 44 to 47. In this loop, the values of $|\psi_j^{nt}|^2 = |\psi(x_j, nt)|^2$ are also computed. Since the matrix elements d_j and c do not change with time, we need to compute the matrix elements α_j and γ_j only once, what we achieve by the query in line 29.

From line 48 to 51 the expectation value $\langle x \rangle_{nt}$ is calculated according to equation (15). For this purpose, we use the trapezoidal rule (see references [2][3]). In fact, we could do without line 48 because we do not allow the wave packet to reach the boundaries and thus we have $|\psi(x_0, nt)|^2 \approx 0$ and $|\psi(x_{j_b}, nt)|^2 \approx 0$. In addition, due to our choice $x_0 = 0$ the first term is identically equal to zero already.

Now we can output the results: wave packet $|\psi(x, nt)|^2$, expectation value $\langle x \rangle_{nt}$ und time nt. After that, the counting index n is increased by 1 (line 52) and an extended query starts on line 53. If NV$= \pm 1$ and the wave packet has not yet come too close to the boundaries and the expectation value of position is in the range of $x_M < \langle x \rangle_{nt} < b - x_M$, the program jumps back to line 28. Otherwise the program is finished. If NV$= 0$ and $n \leq n_T$, the program also jumps back to line 28. Otherwise the wave packet according to equation (59) is calculated separately at time $T = n_T t$ (lines 58 to 62) and then the program is finished.

Examples

The presentation of the output data is best done in graphical form (see figure 2).

The upper diagram shows the probability density $|\psi(x, t)|^2$. The lower diagram shows the potential shape $V(x)$ and the movement of the expectation value $\langle x \rangle_t$. The time is displayed in the upper diagram in units of $1\,\mathrm{bsec} = 10^{-21}\,\mathrm{sec}$. Here we make use of the conversion $1\,\mathrm{fm}\sqrt{\mathrm{u/MeV}} = 0.1018\,\mathrm{bsec}$.

(i) Test of accuracy. Let us consider the case with no potential at all (NV=0). We calculate two wave packets, numerically and analytically, and compare them at time T. If their shapes largely coincide, the values we have chosen for j_b and n_T are sufficiently large for a given energy E. Suitable values are $j_b \approx n_T \approx 1000$ for $E = 10\,\mathrm{MeV}$ and $j_b \approx n_T \approx 2000$ for $E = 40\,\mathrm{MeV}$.

(ii) Partial reflection. Partial reflection at the potential boundaries occurs both at the potential well for arbitrary energies E (see figure 3), and at the potential barrier for $E > E_{\mathrm{pot}}$ (see figure 4).

Strictly speaking, we have to deal not only with *one* but with *multiple* reflections. The first partial reflection occurring at the left boundary is the most striking one. Then there is another significant reflection occurring at the right boundary. These partial reflections at the potential boundaries are going on as long as the wave packet is partly trapped between these boundaries.



Figure 2: Starting position of the wave packet ($E = 25\,\mathrm{MeV}$, $x_V = 10\,\mathrm{fm}$)

The superposition of reflected and non-reflected parts of the wave packet results in interesting patterns of interference. The distance between two maxima or two minima is exactly equal to half of the wavelength of the underlying "carrier wave" ψ. Please note that the wavelength depends on the difference $E - V(x)$ and that the maxima and minima stay put and do not move.

(iii) Tunneling. This effect can be found in the case of a potential barrier with $E < E_{\mathrm{pot}}$ (see figure 5).

The wave packet partly penetrates into the potential barrier at the left boundary and then escapes from the barrier over the right boundary. Again, the patterns of interference mentioned in (ii) show up.

(iv) Resonance. This effect can be found in the case of a potential barrier with $E \approx E_{\mathrm{pot}}$ (see figure 6).

The wave packet is partly trapped between the potential boundaries, moving back and forth very slowly. Due to partial reflections, it can take quite some time until the wave packet vanishes.



Figure 3: Partial reflection at the potential well ($E = 10\,\mathrm{MeV}$, $x_V = 7\,\mathrm{fm}$)



Figure 4: Partial reflection at the potential barrier ($E = 40\,\mathrm{MeV}$, $x_V = 7\,\mathrm{fm}$)



Figure 5: Tunneling ($E = 15\,\mathrm{MeV}$, $x_V = 1.2\,\mathrm{fm}$)



Figure 6: Resonance ($E = 20\,\mathrm{MeV}$, $x_V = 7\,\mathrm{fm}$)

References

[1] A. Goldberg, H.M. Schey, J.L. Schwartz: American Journal of Physics **35**, 177 (1967)

[2] E.W. Schmid, G. Spitz, W. Lösch: *Physikalische Simulationen mit dem Personalcomputer,* 2. Auflage, Springer (1993)

[3] E.W. Schmid, G. Spitz, W. Lösch: *Theoretical Physics on the Personal Computer,* 2nd Edition, Springer (1990)

[4] F.R. Gantmacher: *Matrizentheorie,* Springer (1986)